contract EaselyStandardCollection is ERC721Enumerable, Ownable, IEaselyContractInitializer {    
    using Strings for uint256;

    /**
     * @dev Auction structure that includes:
     * 
     * @param address topBidder - Current top bidder who has already paid the price param below. Is
     *        initialized with address(0) when there have been no bids. When a bidder gets outBid,
     *        the old topBidder will get the price they paid returned.
     * @param uint256 price - Current top price paid by the topBidder.
     * @param uint256 startTimestamp - When the auction can start getting bidded on.
     * @param uint256 endTimestamp - When the auction can no longer get bid on.
     * @param uint256 minBidIncrement - The minimum each new bid has to be greater than the previous
     *        bid in order to be the next topBidder.
     * @param uint256 minLastBidDuration - The minimum time each bid must hold the highest price before
     *        the auction can settle. If people keep bidding, the auction can last for much longer than 
     *        the initial endTimestamp, and endTimestamp will continually be updated.
     */
    struct Auction {
        address topBidder;
        uint256 price;
        uint256 startTimestamp;
        uint256 endTimestamp;
        uint256 minBidIncrement;
        uint256 minLastBidDuration;
    }

    /* Determines if every token in this contract is burnable/claimable by default */
    bool public burnable;
    bool private hasInit = false;

    /* see {IEaselyPayout} for more */
    address public constant payoutContractAddress = 0x68f5C1e24677Ac4ae845Dde07504EAaD98f82572;
    /* Optional addresses to distribute royalties for primary sales of this collection */
    address[] public royalties;
    /* Optional basis points for above royalties addresses for primary sales of this collection */
    uint256[] public royaltiesBPS;
    /* Optional basis points for the owner for secondary sales of this collection */
    uint256 public secondaryOwnerBPS;
    uint256 private nextTokenId = 0;
    /* Optional basis points for the owner for secondary sales of this collection */
    uint256 public timePerDecrement = 300;
    uint256 public constant maxRoyaltiesBPS = 9500;
    uint256 public constant maxSecondaryBPS = 1000;

    /* Mapping if a tokenId has an active auction or not */
    mapping(uint256 => Auction) private _tokenIdToAuction;
    /* Mapping if a tokenId can be claimed */
    mapping(uint256 => bool) private _tokenIdIsClaimable;
    /* Mapping for what the generated token's URI is if the token is claim */
    mapping(uint256 => string) private _tokenIdToPostClaimURI;
    /* Mapping to the active version for all signed transactions */
    mapping(address => uint256) private _addressToActiveVersion;
    /* Cancelled or finalized sales by hash to determine buyabliity */
    mapping(bytes32 => bool) private _cancelledOrFinalizedSales;

    event AuctionCreated(uint256 tokenId, uint256 startingPrice, uint256 startingTimestamp, uint256 endingTimestamp, uint256 minBidIncrement, uint256 minLastBidDuration, address seller);
    event AuctionEndTimeAltered(uint256 tokenId, uint256 endTime, address seller);
    event AuctionCancelled(uint256 tokenId, address seller);
    event AuctionBidded(uint256 tokenId, uint256 newPrice, address bidder);
    event AuctionSettled(uint256 tokenId, uint256 price, address buyer, address seller);

    event ClaimableClaimed(address claimer, uint256 originalTokenId, uint256 newTokenId, string originalIpfs);
    event SaleCancelled(bytes32 hash);
    event SaleCompleted(bytes32 hash);

    /**
     * @dev Constructor function
     */
    constructor(
        bool[8] memory bools, address[8] memory addresses, uint256[32] memory uints, string[8] memory strings
    ) ERC721(strings[0], strings[1]) {
        addresses[0] = _msgSender();
        _init(bools, addresses, uints, strings);
    }

    function init(
        bool[8] memory bools, 
        address[8] memory addresses, 
        uint256[32] memory uints, 
        string[8] memory strings
    ) external override {
        _init(bools, addresses, uints, strings);
    }

    function _init(
        bool[8] memory bools, 
        address[8] memory addresses, 
        uint256[32] memory uints, 
        string[8] memory strings
    ) internal {
        require(!hasInit, "Already has be initiated");
        hasInit = true;
        burnable = bools[0];
        
        _owner = addresses[0];
        address[4] memory royaltiesAddrs = [addresses[1], addresses[2], addresses[3], addresses[4]];
        // Only used for local testing.
        // payoutContractAddress = addresses[5];

        _name = strings[0];
        _symbol = strings[1];

        _setSecondary(uints[0]);
        _setRoyalties(royaltiesAddrs, [uints[1], uints[2], uints[3], uints[4]]);
        if (uints[5] != 0) {
            timePerDecrement = uints[5];
        }
    }

    /**
     * @dev Sets secondary BPS amount
     */
    function _setSecondary(uint256 secondary) internal {
        secondaryOwnerBPS = secondary;
        require(secondaryOwnerBPS <= maxSecondaryBPS, "Cannot take more than 10% of secondaries");
    }

    /**
     * @dev Sets primary royalties
     */
    function _setRoyalties(address[4] memory newRoyalties, uint256[4] memory bps) internal {
        require(bps[0] + bps[1] + bps[2] + bps[3] <= maxRoyaltiesBPS, "Royalties too high");
        royalties = newRoyalties;
        royaltiesBPS = bps;
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual override returns (string memory) {
        return "ipfs://";
    }

    /**
     * @dev Changing _beforeTokenTransfer to lock tokens that are in an auction so
     * that owner cannot transfer the token as people are bidding on it.
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual override { 
        require(!_validAuction(_tokenIdToAuction[tokenId]), "Cannot transfer a token in an auction");

        super._beforeTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev checks if a token is in an auction or not. We make sure that no active auction can 
     * have an endTimestamp of 0.
     */
    function _validAuction(Auction memory auction) internal pure returns (bool) {
        return auction.endTimestamp != 0;
    }

    /**
     * @dev helper method get ownerRoyalties into an array form
     */
    function _ownerRoyalties() internal view returns (address[] memory) {
        address[] memory ownerRoyalties = new address[](1);
        ownerRoyalties[0] = owner();
        return ownerRoyalties;
    }

    /**
     * @dev helper method get secondary BPS into array form
     */
    function _ownerBPS() internal view returns (uint256[] memory) {
        uint256[] memory ownerBPS = new uint256[](1);
        ownerBPS[0] = secondaryOwnerBPS;
        return ownerBPS;
    }

    /**
     * @dev Hash an order, returning the hash that a client must sign, including the standard message prefix
     * @return Hash of message prefix and order hash per Ethereum format
     */
    function _hashToCheck(
        address owner,
        uint256 version,
        uint256 tokenId,
        uint256[4] memory pricesAndTimestamps,
        string memory ipfsHash,
        string memory claimedIpfsHash,
        bool claimable
    ) internal view returns (bytes32) {
        return _toEthSignedMessageHash(_hash(owner, version, tokenId, pricesAndTimestamps, ipfsHash, claimedIpfsHash, claimable));
    }

    /**
     * @dev First checks if a sale is valid by checking that the hash has not been cancelled or already completed
     * and that the correct address has given the signature. If both checks pass we mark the hash as complete and
     * emit an event.
     */
    function _markHashForSale(
        address owner,
        uint256 version,
        uint256 tokenId,
        uint256[4] memory pricesAndTimestamps,
        string memory ipfsHash,
        string memory claimedIpfsHash,
        bool claimable,
        uint8 v, 
        bytes32 r,
        bytes32 s
    ) internal {
        bytes32 hash = _hashToCheck(owner, version, tokenId, pricesAndTimestamps, ipfsHash, claimedIpfsHash, claimable);
        require(!_cancelledOrFinalizedSales[hash], "Sale no longer active");
        require(ecrecover(hash, v, r, s) == owner, "Not signed by current token owner");
        _cancelledOrFinalizedSales[hash] = true;

        emit SaleCompleted(hash);
    }

    /**
     * @dev Hash an order, returning the hash that a client must sign, including the standard message prefix
     * @return Hash of message prefix and order hash per Ethereum format
     */
    function _hash(
        address owner,
        uint256 version,
        uint256 tokenId,
        uint256[4] memory pricesAndTimestamps,
        string memory ipfsHash,
        string memory claimedIpfsHash,
        bool claimable
    ) internal view returns (bytes32) {
        return keccak256(abi.encode(address(this), block.chainid, owner, version, tokenId, pricesAndTimestamps, ipfsHash, claimedIpfsHash, claimable));
    }

    /**
     * @dev Current price for a sale which is calculated for the case of a descending auction. So
     * the ending price must be less than the starting price and the auction must have already started.
     * Standard single fare sales will have a matching starting and ending price.
     */
    function _currentPrice(uint256[4] memory pricesAndTimestamps) internal view returns (uint256) {
        uint256 startingPrice = pricesAndTimestamps[0];
        uint256 endingPrice = pricesAndTimestamps[1];
        uint256 startingTimestamp = pricesAndTimestamps[2];
        uint256 endingTimestamp = pricesAndTimestamps[3];

        uint256 currTime = block.timestamp;
        require(currTime >= startingTimestamp, "Has not started yet");
        require(startingTimestamp < endingTimestamp, "Must end after it starts");
        require(startingPrice >= endingPrice, "Ending price cannot be bigger");

        if (startingPrice == endingPrice || currTime > endingTimestamp) {
            return endingPrice;
        }

        uint256 diff = startingPrice - endingPrice;
        uint256 decrements = (currTime - startingTimestamp) / timePerDecrement;

        // This cannot equal 0 because if endingTimestamp == startingTimestamp, requirements will fail
        uint256 totalDecrements = (endingTimestamp - startingTimestamp) / timePerDecrement;

        return startingPrice - diff / totalDecrements * decrements;
    }

    /**
     * toEthSignedMessageHash
     * @dev prefix a bytes32 value with "\x19Ethereum Signed Message:"
     * and hash the result
     */
    function _toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n32", hash));
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC721Enumerable).interfaceId 
            || interfaceId == type(Ownable).interfaceId
            || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns the current auction variables for a tokenId if the auction is present
     */
    function getAuction(uint256 tokenId) external view returns (Auction memory) {
        require(_validAuction(_tokenIdToAuction[tokenId]), "This auction does not exist");
        
        return _tokenIdToAuction[tokenId];
    }

    /**
     * @dev Returns the current activeVersion of an address both used to create signatures
     * and to verify signatures of {buyExistingToken} and {buyNewToken}
     */
    function getActiveVersion(address address_) external view returns (uint256) {
        return _addressToActiveVersion[address_];
    }

    /** 
     * @dev See {_currentPrice}
     */
    function getCurrentPrice(uint256[4] memory pricesAndTimestamps) external view returns (uint256) {
        return _currentPrice(pricesAndTimestamps);
    }

    /**
     * @dev Usable by the owner of any token initiate a sale for their token. This does not
     * lock the tokenId and the owner can freely trade their token because unlike auctions
     * sales would be immediate.
     */
    function hashToSignToSellToken(
        uint256 version,
        uint256 tokenId,
        uint256[4] memory pricesAndTimestamps
    ) external view returns (bytes32) {
        require(_msgSender() == ownerOf(tokenId), "Not the owner of the token");
        return _hash(_msgSender(), version, tokenId, pricesAndTimestamps, "", "", false);
    }

    /**
     * @dev Usable by the owner of this collection to sell a new token. The owner can decide what
     * the tokenURI of it will be and if the token is claimable and what the claimable hash would be
     */
    function hashToSignToSellNewToken(
        bool claimable,
        uint256 version,
        uint256[4] memory pricesAndTimestamps,
        string memory ipfsHash,
        string memory claimedIpfsHash
    ) external view onlyOwner returns (bytes32) {
        require(bytes(ipfsHash).length > 0, "Invalid ipfsHash");
        return _hash(_msgSender(), version, 0, pricesAndTimestamps, ipfsHash, claimedIpfsHash, claimable);
    }

    /**
     * @dev With a hash signed by the method {hashToSignToSellToken} any user sending enough value can buy
     * the token from the seller. These are all considered secondary sales and will give a cut to the 
     * owner of the contract based on the secondaryOwnerBPS.
     */
    function buyExistingToken(
        address seller,
        uint256 version,
        uint256 tokenId,
        uint256[4] memory pricesAndTimestamps,
        uint8 v, 
        bytes32 r,
        bytes32 s
    ) external payable {
        uint256 currentPrice = _currentPrice(pricesAndTimestamps);

        require(_addressToActiveVersion[seller] == version, "Incorrect signature version");
        require(msg.value >= currentPrice, "Not enough ETH to buy");

        _markHashForSale(seller, version, tokenId, pricesAndTimestamps, "", "", false, v, r, s);

        _transfer(seller, _msgSender(), tokenId);

        IEaselyPayout(payoutContractAddress).splitPayable{ value: currentPrice }(seller, _ownerRoyalties(), _ownerBPS());
        payable(_msgSender()).transfer(msg.value - currentPrice);
    }

    /**
     * @dev With a hash signed by the method {hashToSignToSellNewToken} any user sending enough value can
     * mint the token from the contract. These are all considered primary sales and will give a cut to the 
     * royalties defined in the contract.
     */
    function buyNewToken(
        bool claimable,
        uint256 version,
        uint256[4] memory pricesAndTimestamps,
        string memory ipfsHash,
        string memory claimedIpfsHash,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external payable {
        uint256 currentPrice = _currentPrice(pricesAndTimestamps);

        require(_addressToActiveVersion[owner()] == version, "Incorrect signature version");
        require(msg.value >= currentPrice, "Not enough ETH to buy");

        _markHashForSale(owner(), version, 0, pricesAndTimestamps, ipfsHash, claimedIpfsHash, claimable, v, r, s);

        _safeMint(_msgSender(), nextTokenId);
        _setTokenURI(nextTokenId, ipfsHash);
        
        if (claimable) {
            _tokenIdIsClaimable[nextTokenId] = true;
            _tokenIdToPostClaimURI[nextTokenId] = claimedIpfsHash;
        }
        
        nextTokenId = nextTokenId + 1;

        IEaselyPayout(payoutContractAddress).splitPayable{ value: currentPrice }(owner(), royalties, royaltiesBPS);
        payable(_msgSender()).transfer(msg.value - currentPrice);
    }

    /**
     * @dev Usable to cancel hashes generated from both {hashToSignToSellNewToken} and {hashToSignToSellToken}
     */
    function cancelSale(
        bool claimable,
        uint256 version,
        uint256 tokenId,
        uint256[4] memory pricesAndTimestamps,
        string memory ipfsHash,
        string memory claimedIpfsHash
    ) external {
        bytes32 hash = _hashToCheck(_msgSender(), version, tokenId, pricesAndTimestamps, ipfsHash, claimedIpfsHash, claimable);
        _cancelledOrFinalizedSales[hash] = true;
        emit SaleCancelled(hash);
    }

    /**
     * @dev Usable by any user to update the version that they want their signatures to check. This is helpful if
     * an address wants to mass invalidate their signatures without having to call cancelSale on each one.
     */
    function updateVersion(uint256 version) external {
        _addressToActiveVersion[_msgSender()] = version;
    }

    /**
     * @dev For any user who owns a token that is claimable. If the token has an
     * associated post claim hash then the claimer will get a newly minted token
     * with that hash after claiming. 
     *
     * Claimed tokens that refer to off-chain benefits will be facillitated
     * by Easely, but are the responsibility of the contract creator to 
     * deliver on the promises.
     */
    function claimToken(uint256 tokenId) external {
        require(_exists(tokenId), "tokenId must exist");
        require(_tokenIdIsClaimable[tokenId] || burnable, "tokenId must be claimable");
        require(ownerOf(tokenId) == _msgSender(), "Only current tokenOwner can claim");

        // lock the token from being claimed and thus also from being transferred
        _tokenIdIsClaimable[tokenId] = false;

        // If URI is set, mint the tagged token
        if (bytes(_tokenIdToPostClaimURI[tokenId]).length > 0) {
            _safeMint(_msgSender(), nextTokenId);
            _setTokenURI(nextTokenId, _tokenIdToPostClaimURI[tokenId]);

            emit ClaimableClaimed(_msgSender(), tokenId, nextTokenId, tokenURI(tokenId));
            nextTokenId = nextTokenId + 1;
        }

        _burn(tokenId);
    }

    /**
     * @dev Creates an auction for a token and locks it from being transferred until the auction ends
     * the auction can end if the endTimestamp has been reached and can be cancelled prematurely if
     * there has been no bids yet.
     *
     * @param tokenId uint256 for the token to put on auction. Must exist and be on the auction already
     * @param startingPrice uint256 for the starting price an interested owner must bid
     * @param startingTimestamp uint256 for when the auction can start taking bids
     * @param endingTimestamp uint256 for when the auction has concluded and can no longer take bids
     * @param minBidIncrement uint256 the minimum each interested owner must bid over the latest bid
     * @param minLastBidDuration uint256 the minimum time a bid needs to be live before the auction can end.
     *        this means that an auction can extend past its original endingTimestamp
     */
    function createAuction(
        uint256 tokenId,
        uint256 startingPrice,
        uint256 startingTimestamp,
        uint256 endingTimestamp,
        uint256 minBidIncrement,
        uint256 minLastBidDuration
    ) external {
        require(endingTimestamp > block.timestamp, "Cannot create an auction in the past");
        require(!_validAuction(_tokenIdToAuction[tokenId]), "Token is already on auction");
        require(minBidIncrement > 0, "Min bid must be a positive number");
        require(_msgSender() == ownerOf(tokenId), "Must own token to create auction");

        Auction memory auction = Auction(address(0), startingPrice, startingTimestamp, endingTimestamp, minBidIncrement, minLastBidDuration);

        // This locks the token from being sold
        _tokenIdToAuction[tokenId] = auction;
        emit AuctionCreated(tokenId, startingPrice, startingTimestamp, endingTimestamp, minBidIncrement, minLastBidDuration, ownerOf(tokenId));
    }

    /** 
     * @dev Lets the token owner alter the end time of an auction in case they want to end an auction early or extend
     * the auction. This can only be called when the auction has not yet been concluded and is not within 
     * a minLastBidDuration from concluding.
     */
    function alterEndTime(uint256 tokenId, uint256 endTime) external {
        // 0 EndTimestamp is reserved to check if a tokenId is on auction or not
        require(endTime != 0, "End time cannot be 0");
        require(_msgSender() == ownerOf(tokenId), "Only token owner can alter end time");
        Auction memory auction = _tokenIdToAuction[tokenId];
        require(auction.endTimestamp > block.timestamp + auction.minLastBidDuration, "Auction has already ended");

        auction.endTimestamp = endTime;

        _tokenIdToAuction[tokenId] = auction;
        emit AuctionEndTimeAltered(tokenId, endTime, ownerOf(tokenId));
    }

    /**
     * @dev Allows the token owner to cancel an auction that does not yet have a bid.
     */
    function cancelAuction(uint256 tokenId) external {
        require(_msgSender() == ownerOf(tokenId), "Only token owner can cancel auction");
        Auction memory auction = _tokenIdToAuction[tokenId];
        require(auction.topBidder == address(0), "Cannot cancel an auction with a bid");

        delete _tokenIdToAuction[tokenId];
        emit AuctionCancelled(tokenId, ownerOf(tokenId));
    }

    /**
     * @dev Method that anyone can call to settle the auction. It is available to everyone
     * because the settlement is not dependent on the message sender, and will allow either
     * the buyer, the seller, or a third party to cover the gas fees to settle. The burdern of
     * the auction to settle should be on the seller, but in case there are issues with
     * the seller settling we will not be locked from settling.
     *
     * If the seller is the contract owner, this is considered a primary sale and royalties will
     * be paid to primiary royalties. If the seller is a user then it is a secondary sale and
     * the contract owner will get a secondary sale cut.
     */
    function settleAuction(uint256 tokenId) external {
        Auction memory auction = _tokenIdToAuction[tokenId];
        address tokenOwner = ownerOf(tokenId);
        require(block.timestamp > auction.endTimestamp, "Auction must end to be settled");
        require(auction.topBidder != address(0), "No bidder, cancel the auction instead");

        // This will allow transfers again
        delete _tokenIdToAuction[tokenId];

        _transfer(tokenOwner, auction.topBidder, tokenId);
    
        if (tokenOwner == owner()) {
            IEaselyPayout(payoutContractAddress).splitPayable{ value: auction.price }(tokenOwner, royalties, royaltiesBPS);
        } else {
            address[] memory ownerRoyalties = new address[](1);
            uint256[] memory ownerBPS = new uint256[](1);
            ownerRoyalties[0] = owner();
            ownerBPS[0] = secondaryOwnerBPS;

            IEaselyPayout(payoutContractAddress).splitPayable{ value: auction.price }(tokenOwner, ownerRoyalties, ownerBPS);
        }

        emit AuctionSettled(tokenId, auction.price, auction.topBidder, tokenOwner);
    }

    /**
     * @dev Allows any potential buyer to submit a bid on a token with an auction. When outbidding the current topBidder
     * the contract returns the value that the previous bidder had escrowed to the contract.
     */
    function bidOnAuction(uint256 tokenId) external payable {
        uint256 timestamp = block.timestamp;
        Auction memory auction = _tokenIdToAuction[tokenId];
        uint256 msgValue = msg.value;

        // Tokens that are not on auction always have an endTimestamp of 0
        require(timestamp <= auction.endTimestamp, "Auction has already ended");
        require(timestamp >= auction.startTimestamp, "Auction has not started yet");

        uint256 minPrice = auction.price + auction.minBidIncrement;
        if (auction.topBidder == address(0)) {
            minPrice = auction.price;
        }
        require(msgValue >= minPrice, "Bid is too small");

        uint256 endTime = auction.endTimestamp;
        if (endTime < auction.minLastBidDuration + timestamp) {
            endTime = timestamp + auction.minLastBidDuration;
        }

        Auction memory newAuction = Auction(_msgSender(), msgValue, auction.startTimestamp, endTime, auction.minBidIncrement, auction.minLastBidDuration);

        if (auction.topBidder != address(0)) {
            // Give the old top bidder their money back
            payable(auction.topBidder).transfer(auction.price);
        }

        _tokenIdToAuction[tokenId] = newAuction;
        emit AuctionBidded(tokenId, newAuction.price, newAuction.topBidder);
    }

    /**
     * @dev see {_setRoyalties}
     */
    function setRoyalties(address[4] memory newRoyalties, uint256[4] memory bps) external onlyOwner {
        _setRoyalties(newRoyalties, bps);
    }

    /**
     * @dev see {_setSecondary}
     */
    function setSecondaryBPS(uint256 bps) external onlyOwner() {
        _setSecondary(bps);
    }

    /**
     * @dev Allows the owner to create a new token with ipfsHash as the tokenURI.
     */
    function mint(address collectorAddress, string memory ipfsHash) external onlyOwner {
        // mint token
        _safeMint(collectorAddress, nextTokenId);
        _setTokenURI(nextTokenId, ipfsHash);
        
        nextTokenId = nextTokenId + 1;
    }

    /**
     * @dev Adds a claimable hash to an existing token. If claimedIpfsHash is exactly "" then
     * no token will be created when the token is claimed, otherwise a new token with that hash 
     * will be given to the token owner when they call {claimToken}
     */
    function addClaimable(
        uint256 tokenId,
        string memory claimedIpfsHash
    ) external onlyOwner {
        require(_exists(tokenId), "tokenId must exist");
        require(!_tokenIdIsClaimable[tokenId], "Claimable already exists");

        _tokenIdIsClaimable[tokenId] = true;
        _tokenIdToPostClaimURI[tokenId] = claimedIpfsHash;
    }

    /**
     * @dev Allows the owner to create a new token with ipfsHash as the tokenURI that is also claimable. 
     * If claimedIpfsHash is exactly "" then no token will be created when the token is claimed, otherwise 
     * a new token with hash claimedIpfsHash will be given to the token owner when they call {claimToken}
     */
    function mintClaimable(
        address collectorAddress, 
        string memory ipfsHash,
        string memory claimedIpfsHash
    ) external onlyOwner {
        // mint token
        _safeMint(collectorAddress, nextTokenId);
        _setTokenURI(nextTokenId, ipfsHash);

        _tokenIdIsClaimable[nextTokenId] = true;
        _tokenIdToPostClaimURI[nextTokenId] = claimedIpfsHash;
        
        nextTokenId = nextTokenId + 1;
    }
}
/**
 * @dev This implements three things on top of the standard ERC extensions.
 * 1. The ability for contract owners to mint claimable (burnable) tokens that can optionally
 *    generate another token.
 * 2. The ability for current token owners to create ascending auctions, which locks the token for
 *    the duration of the auction.
 * 3. The ability for current token owners to lazily sell their tokens in this contract instead of 
 *    needing a marketplace contract.
 */
function mint(address collectorAddress, string memory ipfsHash) external onlyOwner {
        // mint token
        _safeMint(collectorAddress, nextTokenId);
        _setTokenURI(nextTokenId, ipfsHash);
        
        nextTokenId = nextTokenId + 1;
    }
/**
 * @dev Allows the owner to create a new token with ipfsHash as the tokenURI.
 */